

NAG Library Routine Document

G01NBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G01NBF computes the moments of ratios of quadratic forms in Normal variables and related statistics.

2 Specification

```

SUBROUTINE G01NBF (CASE, MEAN, N, A, LDA, B, LDB, C, LDC, ELA, EMU,      &
                  SIGMA, LDSIG, L1, L2, LMAX, RMOM, ABSERR, EPS, WK,    &
                  IFAIL)
INTEGER          N, LDA, LDB, LDC, LDSIG, L1, L2, LMAX, IFAIL
REAL (KIND=nag_wp) A(LDA,N), B(LDB,N), C(LDC,*), ELA(*), EMU(*),      &
                  SIGMA(LDSIG,N), RMOM(L2-L1+1), ABSERR, EPS,        &
                  WK(3*N*N+(8+L2)*N)
CHARACTER(1)    CASE, MEAN

```

3 Description

Let x have an n -dimensional multivariate Normal distribution with mean μ and variance-covariance matrix Σ . Then for a symmetric matrix A and symmetric positive semidefinite matrix B , G01NBF computes a subset, l_1 to l_2 , of the first 12 moments of the ratio of quadratic forms

$$R = x^T A x / x^T B x.$$

The s th moment (about the origin) is defined as

$$E(R^s), \tag{1}$$

where E denotes the expectation. Alternatively, this routine will compute the following expectations:

$$E(R^s (a^T x)) \tag{2}$$

and

$$E(R^s (x^T C x)), \tag{3}$$

where a is a vector of length n and C is a n by n symmetric matrix, if they exist. In the case of (2) the moments are zero if $\mu = 0$.

The conditions of theorems 1, 2 and 3 of Magnus (1986) and Magnus (1990) are used to check for the existence of the moments. If all the requested moments do not exist, the computations are carried out for those moments that are requested up to the maximum that exist, l_{MAX} .

This routine is based on the routine QRMOM written by Magnus and Pesaran (1993a) and based on the theory given by Magnus (1986) and Magnus (1990). The computation of the moments requires first the computation of the eigenvectors of the matrix $L^T B L$, where $LL^T = \Sigma$. The matrix $L^T B L$ must be positive semidefinite and not null. Given the eigenvectors of this matrix, a function which has to be integrated over the range zero to infinity can be computed. This integration is performed using D01AMF.

4 References

Magnus J R (1986) The exact moments of a ratio of quadratic forms in Normal variables *Ann. Üconom. Statist.* **4** 95–109

Magnus J R (1990) On certain moments relating to quadratic forms in Normal variables: Further results *Sankhyā, Ser. B* **52** 1–13

Magnus J R and Pesaran B (1993a) The evaluation of cumulants and moments of quadratic forms in Normal variables (CUM): Technical description *Comput. Statist.* **8** 39–45

Magnus J R and Pesaran B (1993b) The evaluation of moments of quadratic forms and ratios of quadratic forms in Normal variables: Background, motivation and examples *Comput. Statist.* **8** 47–55

5 Arguments

- 1: CASE – CHARACTER(1) *Input*
On entry: indicates the moments of which function are to be computed.
CASE = 'R' (Ratio)
 $E(R^s)$ is computed.
CASE = 'L' (Linear with ratio)
 $E(R^s(a^T x))$ is computed.
CASE = 'Q' (Quadratic with ratio)
 $E(R^s(x^T C x))$ is computed.
Constraint: CASE = 'R', 'L' or 'Q'.
- 2: MEAN – CHARACTER(1) *Input*
On entry: indicates if the mean, μ , is zero.
MEAN = 'Z'
 μ is zero.
MEAN = 'M'
The value of μ is supplied in EMU.
Constraint: MEAN = 'Z' or 'M'.
- 3: N – INTEGER *Input*
On entry: n , the dimension of the quadratic form.
Constraint: $N > 1$.
- 4: A(LDA, N) – REAL (KIND=nag_wp) array *Input*
On entry: the n by n symmetric matrix A . Only the lower triangle is referenced.
- 5: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which G01NBF is called.
Constraint: $LDA \geq N$.
- 6: B(LDB, N) – REAL (KIND=nag_wp) array *Input*
On entry: the n by n positive semidefinite symmetric matrix B . Only the lower triangle is referenced.
Constraint: the matrix B must be positive semidefinite.

- 7: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which G01NBF is called.
Constraint: $LDB \geq N$.
- 8: C(LDC,*) – REAL (KIND=nag_wp) array *Input*
Note: the second dimension of the array C must be at least N if CASE = 'Q', and at least 1 otherwise.
On entry: if CASE = 'Q', C must contain the n by n symmetric matrix C ; only the lower triangle is referenced.
 If CASE \neq 'Q', C is not referenced.
- 9: LDC – INTEGER *Input*
On entry: the first dimension of the array C as declared in the (sub)program from which G01NBF is called.
Constraints:
 if CASE = 'Q', $LDC \geq N$;
 otherwise $LDC \geq 1$.
- 10: ELA(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array ELA must be at least N if CASE = 'L', and at least 1 otherwise.
On entry: if CASE = 'L', ELA must contain the vector a of length n , otherwise ELA is not referenced.
- 11: EMU(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array EMU must be at least N if MEAN = 'M', and at least 1 otherwise.
On entry: if MEAN = 'M', EMU must contain the n elements of the vector μ .
 If MEAN = 'Z', EMU is not referenced.
- 12: SIGMA(LDSIG,N) – REAL (KIND=nag_wp) array *Input*
On entry: the n by n variance-covariance matrix Σ . Only the lower triangle is referenced.
Constraint: the matrix Σ must be positive definite.
- 13: LDSIG – INTEGER *Input*
On entry: the first dimension of the array SIGMA as declared in the (sub)program from which G01NBF is called.
Constraint: $LDSIG \geq N$.
- 14: L1 – INTEGER *Input*
On entry: the first moment to be computed, l_1 .
Constraint: $0 < L1 \leq L2$.
- 15: L2 – INTEGER *Input*
On entry: the last moment to be computed, l_2 .
Constraint: $L1 \leq L2 \leq 12$.

- 16: LMAX – INTEGER Output
On exit: the highest moment computed, l_{MAX} . This will be l_2 if IFAIL = 0 on exit.
- 17: RMOM(L2 – L1 + 1) – REAL (KIND=nag_wp) array Output
On exit: the l_1 to l_{MAX} moments.
- 18: ABSERR – REAL (KIND=nag_wp) Output
On exit: the estimated maximum absolute error in any computed moment.
- 19: EPS – REAL (KIND=nag_wp) Input
On entry: the relative accuracy required for the moments, this value is also used in the checks for the existence of the moments.
 If EPS = 0.0, a value of $\sqrt{\epsilon}$ where ϵ is the *machine precision* used.
Constraint: EPS = 0.0 or EPS \geq *machine precision*.
- 20: WK(3 × N × N + (8 + L2) × N) – REAL (KIND=nag_wp) array Workspace
- 21: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if IFAIL \neq 0 on exit, the recommended value is –1. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or –1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01NBF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

- On entry, $N \leq 1$,
- or LDA < N,
- or LDB < N,
- or LDSIG < N,
- or CASE = 'Q' and LDC < N,
- or CASE \neq 'Q' and LDC < 1,
- or L1 < 1,
- or L1 > L2,
- or L2 > 12,
- or CASE \neq 'R', 'L' or 'Q',
- or MEAN \neq 'M' or 'Z',
- or EPS \neq 0.0 and EPS < *machine precision*.

IFAIL = 2

On entry, Σ is not positive definite,
or B is not positive semidefinite or is null.

IFAIL = 3

None of the required moments can be computed.

IFAIL = 4

The matrix $L^T B L$ is not positive semidefinite or is null.

IFAIL = 5

The computation to compute the eigenvalues required in the calculation of moments has failed to converge: this is an unlikely error exit.

IFAIL = 6

Only some of the required moments have been computed, the highest is given by LMAX.

IFAIL = 7

The required accuracy has not been achieved in the integration. An estimate of the accuracy is returned in ABSERR.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The relative accuracy is specified by EPS and an estimate of the maximum absolute error for all computed moments is returned in ABSERR.

8 Parallelism and Performance

G01NBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example is given by Magnus and Pesaran (1993b) and considers the simple autoregression:

$$y_t = \beta y_{t-1} + u_t, \quad t = 1, 2, \dots, n,$$

where $\{u_t\}$ is a sequence of independent Normal variables with mean zero and variance one, and y_0 is known. The least squares estimate of β , $\hat{\beta}$, is given by

$$\hat{\beta} = \frac{\sum_{t=2}^n y_t y_{t-1}}{\sum_{t=2}^n y_t^2}.$$

Thus $\hat{\beta}$ can be written as a ratio of quadratic forms and its moments computed using G01NBF. The matrix A is given by

$$A(i+1, i) = \frac{1}{2}, \quad i = 1, 2, \dots, n-1;$$

$$A(i, j) = 0, \quad \text{otherwise,}$$

and the matrix B is given by

$$B(i, i) = 1, \quad i = 1, 2, \dots, n-1;$$

$$B(i, j) = 0, \quad \text{otherwise.}$$

The value of Σ can be computed using the relationships

$$\text{var}(y_t) = \beta^2 \text{var}(y_{t-1}) + 1$$

and

$$\text{cov}(y_t y_{t+k}) = \beta \text{cov}(y_t y_{t+k-1})$$

for $k \geq 0$ and $\text{var}(y_1) = 1$.

The values of β , y_0 , n , and the number of moments required are read in and the moments computed and printed.

10.1 Program Text

```

Program g01nbfe

!      G01NBF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
!      Use nag_library, Only: g01nbf, nag_wp
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
!      Real (Kind=nag_wp)         :: abserr, beta, eps, y0
!      Integer                    :: i, ifail, j, l1, l2, lda, ldb, ldc, &
!                                  ldsig, lmax, lwk, n
!
!      .. Local Arrays ..
!      Real (Kind=nag_wp), Allocatable :: a(:,,:), b(:,,:), c(:,,:), ela(:), &
!                                  emu(:), rmom(:), sigma(:,,:), wk(:)
!
!      .. Executable Statements ..
!      Write (nout,*) 'G01NBF Example Program Results'
!      Write (nout,*)
!
!      Skip heading in data file
!      Read (nin,*)

```

```

!      Read in the problem size
      Read (nin,*) beta, y0
      Read (nin,*) n, l1, l2

      lda = n
      ldb = n
      ldc = n
      ldsig = n
      lwk = 3*n*n + (8+12)*n
      Allocate (a(lda,n),b(ldb,n),c(ldc,n),ela(n),emu(n),sigma(ldsig,n),      &
               wk(lwk),rmom(l2-l1+1))

!      Compute A, EMU, and SIGMA for simple autoregression
      Do i = 1, n
        Do j = i, n
          a(j,i) = 0.0E0_nag_wp
          b(j,i) = 0.0E0_nag_wp
        End Do
      End Do
      Do i = 1, n - 1
        a(i+1,i) = 0.5E0_nag_wp
        b(i,i) = 1.0E0_nag_wp
      End Do
      emu(1) = y0*beta
      Do i = 1, n - 1
        emu(i+1) = beta*emu(i)
      End Do
      sigma(1,1) = 1.0E0_nag_wp
      Do i = 2, n
        sigma(i,i) = beta*beta*sigma(i-1,i-1) + 1.0E0_nag_wp
      End Do
      Do i = 1, n
        Do j = i + 1, n
          sigma(j,i) = beta*sigma(j-1,i)
        End Do
      End Do

!      Use default accuracy
      eps = 0.0E0_nag_wp

!      Compute moments
      ifail = -1
      Call g01nbf('Ratio', 'Mean', n, a, lda, b, ldb, c, ldc, ela, emu, sigma, ldsig, l1,      &
                l2, lmax, rmom, abserr, eps, wk, ifail)
      If (ifail/=0) Then
        If (ifail<6) Then
          Go To 100
        End If
      End If

!      Display results
      Write (nout,99999) ' N = ', n, ' BETA = ', beta, ' Y0 = ', y0
      Write (nout,*)
      Write (nout,*) '           Moments'
      Write (nout,*)
      j = 0
      Do i = l1, lmax
        j = j + 1
        Write (nout,99998) i, rmom(j)
      End Do

100    Continue

99999 Format (A,I3,2(A,F6.3))
99998 Format (I3,E12.3)
      End Program g01nbf

```

10.2 Program Data

```
G01NBF Example Program Data
0.8 1.0      : Beta Y0
10  1   3   : N  L1  L1
```

10.3 Program Results

```
G01NBF Example Program Results
```

```
N = 10 BETA = 0.800 Y0 = 1.000
```

```
      Moments
```

```
 1  0.682E+00
 2  0.536E+00
 3  0.443E+00
```
